

# landscape.rem

=====

*A landscape program*

=====

using graphics

```
add the sun on 'the_layer' :
  'sun' : a YELLOW circle of radius 40 at {800, 100}
  place 'sun' in 'the_layer'

add the mountains on 'the_layer' :
  repeat 20 times
    'mountain' : a DARK-GREY mountain with BLACK edges
                  ... of height 300 based at 600
    place 'mountain' in 'the_layer'

add the sea on 'the_layer' :
  'sea' : a DARK-BLUE box of STD-WIDTH x 200 at {
    STD-WIDTH ÷ 2
    STD-HEIGHT - 100
  }
  place 'sea' in 'the_layer'

add the islands on 'the_layer' :
  for each 'location' from 650 to 680 in steps of 3
    'island' : a GREY mountain with WHITE edges
              ... of height 250 based at 'location'
    place 'island' in 'the_layer'

'n' clouds in 'the_layer' :
  apply
    'cloud' : a cloud
    place ('cloud' parts) in 'the_layer'
    'cloud'
  ... 'n' times

animate 'the_clouds' in 'the_window' :
  animate 'the_window' 600 times at 30 ticks per second
    for each 'cloud' in 'the_clouds'
      move 'cloud'
```

=

*A mountain is an isosceles triangle.*

=

```
a 'colour' mountain with 'outline colour' edges
  ... of height 'max_height' based at 'level' :
  'centre' : random STD-WIDTH
  'height' : random 'max_height' + 20
  'width' : random 150 + 300
  'mountain' : a 'colour' shape from {
    { 0, -'height' }
    { 'width' ÷ 2, 0 }
    { -'width' ÷ 2, 0 }
  } scaled by 1 at { 'centre', 'level' }
  'mountain' outline colour 'outline colour'
  'mountain'
```

=

*A cloud is a composite object comprising of two circles, a rectangle and a line.*

*Each part is added separately to the same window layer using the list*

# landscape.rem

*returned from "MY parts".*

*When animated the "move ME" method changes the positions of all the parts.  
The "move ME" also wraps the cloud around when it passes off the window.*

```
=
a cloud :
  - setting up the object's field values
  'base' : random (STD-HEIGHT - 200)
  'across' : random STD-WIDTH
  'r1' : random 10 + 30
  'r2' : random 10 + 20
  'distance' : 10
  if ('r1' > 'r2') ; bigger circle to the right
    'r1' swap 'r2'

  'x1' : ('across' - 'r1') + 'distance'
  'x2' : ('across' + 'r2') - 'distance'
  'y1' : 'base' - 'r1'
  'y2' : 'base' - 'r2'

  'circle A' : a WHITE circle of radius 'r1' at { 'x1', 'y1' }
  'circle A' outline colour BLACK
  'circle B' : a WHITE circle of radius 'r2' at { 'x2', 'y2' }
  'circle B' outline colour BLACK

  'width' : 'x2' - 'x1'
  'height' : 'y1' - 'y2' + 'r1'
  'position' : {
    ('x1' + 'x2') ÷ 2
    ('y1' + 'r1' + 'y2' + 1) ÷ 2
  }

  'box' : a WHITE box of 'width' x 'height' at 'position'
  'line' : a BLACK line from { 'x1', 'y1' + 'r1' } to { 'x2', 'y1' + 'r1' }
  'line' width 2

  - creating the object
  create
    - these are fields
    'circle A' : 'circle A'
    'circle B' : 'circle B'
    'box' : 'box'
    'line' : 'line'
    'size' : 'circle A' radius + 'circle B' radius +
      ... 'circle B' x-pos - 'circle A' x-pos

    - these are methods
  MY parts :
    { 'circle A', 'circle B', 'box', 'line' }

  move ME :
    'left edge' : 'circle A' x-pos - 'circle A' radius
    if ('left edge' < STD-WIDTH) ; ordinary move right
      for each 'part' in (MY parts)
        change 'part' x-pos by 1
    ... otherwise ; wrap the cloud around the window
      for each 'part' in (MY parts)
        change 'part' x-pos by -(STD-WIDTH + 'size')

  'the window' : the graphics panel
  'the window' colour SKY-BLUE
```

# landscape.rem

```
=====
'the back layer' : 'the window' base layer
'the cloud layer' : 'the window' next layer
'the front layer' : 'the window' next layer
```

```
add the sun on 'the back layer'
add the mountains on 'the back layer'
add the sea on 'the back layer'
add the islands on 'the front layer'
update 'the window'
```

```
'the clouds' : 10 clouds in 'the cloud layer'
animate 'the clouds' in 'the window'
```